ЛАБОРАТОРНАЯ РАБОТА №3

ИНТЕРПОЛЯЦИЯ ТАБЛИЧНО ЗАДАННЫХ ФУНКЦИЙ

Пусть на отрезке $[x_{\min}, x_{\max}]$ задана система точек (узлы интерполяции) x_i и значения неизвестной функции в этих узлах f_i, $i = 1,...,N$. Необходимо построить функцию $F(x)$, принимающую в узлах интерполяции x_i, заданные значения f_i:

$$F(x_i) = f_i,\ i = 1,...,N \tag{1}$$

(условия интерполяции).

Локальная интерполяция. На каждом интервале $[x_i, x_{i+1}], 1 \le i \le N-1$ строится своя функция $F_i(x)$.

При кусочно-линейной интерполяции

$$F_i(x) = a_i + b_i x.$$

Неизвестные коэффициенты a_i и b_i определяются из условий (1) на границах интервала:

$$F_i(x_i) = f_i \quad , F_i(x_{i+1}) = f_{i+1}.$$

При сплайн-интерполяции $F_i(x)$ представляется полиномом третьей степени (кубическим сплайном)

$$S_i(x) = a_i + b_i(x - x_i) + c_i(x - x_i)^2 + d_i(x - x_i)^3, \quad i = 1,2,...,N-1. \tag{2}$$

Из условий интерполяции (1), непрерывности $S(x)$, ее первых (отсутствие изломов) и вторых (отсутствие скачков кривизны) производных в узлах можно получить *4N-6* уравнений для *4N-4* неизвестных:

$$\begin{aligned}
&a_i = f_i, \quad i = 1,2,...,N-1,\\
&f_i + b_i h_i + c_i h_i^2 + d_i h_i^3 = f_{i+1}, \quad i = 1,2,...,N-1,\\
&b_i + 2c_i h_i + 3d_i h_i^2 = b_{i+1}, \quad i = 1,2,...,N-2,\\
&2c_i + 6d_i h_i = 2c_{i+1}, \quad i = 1,2,...,N-2,
\end{aligned} \tag{3}$$

где $h_i = x_{i+1} - x_i$.

Для замыкания системы (3) задаются значения вторых производных (кривизны) на концах отрезка интерполяции:

$$S_1^{(2)}(x_1) = s_1'', \quad S_N^{(2)}(x_N) = s_N''$$

или

$$c_1 = s_1''/2, \quad c_N = s_N''/2. \tag{4}$$

Из (3) получаются уравнения для определения коэффициентов сплайна:

$$d_i = \frac{c_{i+1} - c_i}{3h_i}, \quad i = 1,2,...,N-1, \tag{5}$$

$$b_i = \frac{f_{i+1} - f_i}{h_i} - \frac{h_i}{3}(c_{i+1} + 2c_i), \quad i = 1,2,...,N-1 \tag{6}$$

$$h_{i-1}c_{i-1} + 2(h_{i-1} + h_i)c_i + h_i c_{i+1} = 3\left[\frac{f_{i+1} - f_i}{h_i} - \frac{f_i - f_{i-1}}{h_{i-1}}\right], \quad i = 2,3,...,N-1. \quad (7)$$

Трехдиагональная система уравнений (4), (7) решается методом прогонки. После того как найдены все c_i, из (5), (6) определяются остальные неизвестные.

```
subroutine SPSET(N,X,F,s1,sN,A,B,C,D)
! Коэффициенты кубического сплайна
REAL X(N),F(N),A(N),B(N),C(N),D(N),q(N)
A(1)=0.
B(1)=0.
D(1)=s1/2.
C(1)=1.
D(N)=sN/2.
C(N)=1.
do I=2,N-1
 H1=X(I)-X(I-1)
 H2=X(I+1)-X(I)
 A(I)=H1
 C(I)=2.*(H1+H2)
 B(I)=H2
 D(I)=3.*((F(I+1)-F(I))/H2-(F(I)-F(I-1))/H1)
end do
CALL PROGON(N,A,C,B,D,q) ! решение СЛАУ
do I=1,N-1
  H2=X(I+1)-X(I)
  A(I)=F(I)
  C(I)=q(I)
  D(I)=(q(I+1)-q(I))/(3.*H2)
  B(I)=(F(I+1)-F(I))/H2-H2/3.*(q(I+1)+2.*q(I))
end do
END
```

Здесь «N» – количество узлов таблицы, «X» и «F» – массивы табличных данных x_i и f_i, «s1» и «sN» – граничные значения s''_1 и s''_N, «A», «B», «C» и «D» – массивы коэффициентов сплайна a_i, b_i, c_i и d_i. Используется процедура метода прогонки «PROGON».

Зная наборы коэффициентов (по четыре коэффициента для каждого участка), можно определить значение сплайн-функции по формуле (2) в любой точке.

```
subroutine SPVAL(N,x,a,b,c,d,z,S)
! Значение кубического сплайна S(z) в точке z
dimension x(N),a(N),b(N),c(N),d(N)
i=1
3 IF(x(i+1)>z.OR.i==N-1) GOTO 4
i=i+1
GOTO 3
4 z1=z-x(i)
S=a(i)+b(i)*z1+c(i)*z1**2+d(i)*z1**3
end
```

<u>Глобальная интерполяция.</u> Строится одна функция для всего интервала $[x_{\min}, x_{\max}]$. Примером является интерполяционный полином Лагранжа:

$$L(x) = \sum_{i=1}^{N} f_i \prod_{\substack{j=1 \\ i \neq j}}^{N} \frac{x - x_j}{x_i - x_j}. \quad (8)$$

Для повышения точности интерполяции вместо полного набора узлов часто используют несколько узлов, близких к данной точке x. В этом случае в формуле (8) индексы i и j должны изменяться от $i_н$ (левой границы «окна») до $i_к$ (правой границы «окна»), причем ширина «окна» $l = i_к - i_н$ обычно берется равной 4-5.

```
subroutine Lagrok(N,x,f,z,lok,P)
! Полином Лагранжа P(z)
dimension x(N),f(N)
i2=lok/2;if(i2==0)i2=1;i1=i2 ! границы окна in,ik
do j=1,N;if(z>x(i1).and.i1<N)i1=i1+1;end do
in=i1-i2;if(in<1)in=1;ik=in+lok;if(ik<=N)goto 1
ik=N;in=ik-lok
1 P=0.0
do i=in,ik
R=1.0
do j=in,ik
 if(i==j) cycle
 Q=x(i)-x(j)
 S=z-x(j)
 R=R*S/Q
end do
P=P+R*f(i)
end do
```

end

Здесь «lok» – ширина «окна».

Задание. Написать программу для расчета значения $F(z)$ таблично заданной функции $(x_i, f_i)_{i=1,\ldots,N}$ $(x_1 = x_{\min}, \; x_N = x_{\max})$ в произвольной точке $z \in [x_{\min}, x_{\max}]$ с помощью:

1) кусочно-линейной интерполяции;
2) сплайн-интерполяции;
3) полинома Лагранжа в «оконном» варианте.

Исследовать влияние на максимальную абсолютную погрешность $\Delta = \max|f - F|$ применяемых методов интерполяции:

- числа узлов N при $N = 11, \ldots, 81$ (для $s_1'' = s_N'' = 0$, $l = 5$);
- граничных значений s_1'' и s_N'' 2-й производной сплайна при $s_1'' = s_N'' = -1 \ldots 1$ (для $N = 11$);
- ширины «окна» l полинома Лагранжа при $l = 3 \ldots 7$ (для $N = 11$).

Варианты заданий приведены в таблице:

Вариант	$f(x)$, $[x_{\min}, x_{\max}]$	Вариант	$f(x)$, $[x_{\min}, x_{\max}]$
1	$x^2\cos(2x) + x^3/3$, [-1,1]	2	$\sin^3(5x) + \exp(x/3)$, [0,1]
3	$x^2 - (1+x)/tg^2(x/4)$, [1,2]	4	$\sqrt[3]{x+1} - 1/x$, [2,3]
5	$6/x + 2^x\exp(-x^2)$, [1,3]	6	$x^3\cos^2(x/3) + \lg(x/3)$, [1,2]
7	$\sin(3x^2) - 2x/(x+4)$, [-2,0]	8	$2x^2 - \sin^3(5x)$, [0,1]
9	$\exp(x) + x/(1+x^2)$, [-1,1]	10	$x^2 - \cos^2(\pi x)$, [1,3]
11	$0.1e^x - \sin^2 x + 0.5$, [0,1]	12	$x\cos(x^2 - 2x) + x^2$, [-1,1]
13	$\lg x - (x-1)/x$, [1,4]	14	$x\sin(x/2) - x^4/4$, [-2,1]
15	$\cos^2(3x) - \exp(x/5)$, [0,1]	16	$x^3 - (2+x)/ctg(x/5)$, [2,3]
17	$2/x^2 - 3^{\sqrt{x}}\exp(-x)$, [2,3]	18	$x^2\sin^3(x/5) + \log_2(x/2)$, [3,4]
19	$\sqrt{x}\sin(x^3 + 3x^2) - x$, [1,3]	20	$\cos(2x^3) + 3x/(x-1)$, [2,4]
21	$\cos(2x^3) + x^3/(x^2+5)$, [-2,1]	22	$12\sin(x) - 2^{\sqrt{3x}}sh(-x)$, [2,3]
23	$lgx - (x-1)/x$, [1,3]		

Содержание отчета:

1. Постановка задачи.
2. Краткое описание используемых методов.
3. Текст расчетной программы.
4. Листинг результатов.
5. Графики зависимости погрешностей Δ от N, от s_1'' и от l. Определение практического порядка точности методов (показателя степени α в формуле $\Delta \sim N^{-\alpha}$).

6. Выводы.

Указания. Воспользоваться приведенными выше процедурами методов. Пример главной программы:

```
program Interp
parameter (N=11,Nz=31) ! число узлов и промежут. точек
dimension x(N),f(N),Ye(Nz),P(Nz),S(Nz),R(Nz)
dimension a(N),b(N),c(N),d(N)
! табличная функция
xmin=-1.; xmax=1.; dx=(xmax-xmin)/(N-1)
do i=1,N
 x(i)=xmin+(i-1)*dx
 f(i)=fe(x(i))
end do
s1=0.0; sN=0.0              ! ГУ S''(x1)=s1, S''(xN)=sN
call SPSET(N,x,f,s1,sN,a,b,c,d) ! коэффициенты сплайна
dz=(x(N)-x(1))/(Nz-1)
write(*,90)
do i=1,Nz ! сплайн и полином Лагранжа в промежут. точках
 z=x(1)+(i-1)*dz
! call LININT(N,x,f,z,R(i))
 call SPVAL(N,x,a,b,c,d,z,S(i))
 call Lagrok(N,x,f,z,5,P(i))
 Ye(i)=fe(z); dS=abs(Ye(i)-S(i)); dP=abs(Ye(i)-P(i))
 write(*,'(2x,f8.4,5(1pe12.4))')z,Ye(i),S(i),P(i),dS,dP
end do
print*,' max|f-S|=',maxval(abs(Ye-S))
print*,' max|f-L|=',maxval(abs(Ye-P))
90 format(7x,'z',9x,'f',11x,'S',11x,'L',9x,'|f-S|',7x,'|f-L|')
End
function fe(x)
pi=4*atan(1.)
fe=10.*sin(pi*x)
end
```

Программа «Interp» рассчитывает в промежуточных точках, не совпадающих с узлами, и выводит на экран значения табличной и интерполирующих функций и соответствующие абсолютные погрешности. Используется процедура расчета точного значения табличной функции «fe».

Результаты:

z	f	S	L	|f-S|	|f-L|
-1.0000	8.7423E-07	8.7423E-07	8.7423E-07	0.0000E+00	0.0000E+00
-0.9333	-2.0791E+00	-2.0777E+00	-2.0660E+00	1.4238E-03	1.3127E-02
-0.8667	-4.0674E+00	-4.0666E+00	-4.0595E+00	7.8678E-04	7.8330E-03
-0.8000	-5.8779E+00	-5.8779E+00	-5.8779E+00	0.0000E+00	0.0000E+00
-0.7333	-7.4314E+00	-7.4284E+00	-7.4353E+00	3.0904E-03	3.8409E-03
-0.6667	-8.6603E+00	-8.6576E+00	-8.6633E+00	2.6970E-03	3.0718E-03
-0.6000	-9.5106E+00	-9.5106E+00	-9.5106E+00	0.0000E+00	9.5367E-07
-0.5333	-9.9452E+00	-9.9416E+00	-9.9486E+00	3.5763E-03	3.3474E-03
-0.4667	-9.9452E+00	-9.9416E+00	-9.9478E+00	3.5763E-03	2.6188E-03
-0.4000	-9.5106E+00	-9.5106E+00	-9.5106E+00	0.0000E+00	0.0000E+00
-0.3333	-8.6603E+00	-8.6576E+00	-8.6618E+00	2.6960E-03	1.5745E-03
-0.2667	-7.4314E+00	-7.4284E+00	-7.4326E+00	3.0899E-03	1.1706E-03
-0.2000	-5.8779E+00	-5.8779E+00	-5.8779E+00	0.0000E+00	0.0000E+00
-0.1333	-4.0674E+00	-4.0666E+00	-4.0666E+00	7.8678E-04	7.9823E-04
-0.0667	-2.0791E+00	-2.0777E+00	-2.0784E+00	1.4236E-03	7.2718E-04
0.0000	1.6385E-06	1.6374E-06	1.6410E-06	1.0622E-09	2.5743E-09
0.0667	2.0791E+00	2.0777E+00	2.0820E+00	1.4238E-03	2.8675E-03
0.1333	4.0674E+00	4.0666E+00	4.0697E+00	7.8678E-04	2.3465E-03
0.2000	5.8779E+00	5.8779E+00	5.8779E+00	4.7684E-07	4.7684E-07
0.2667	7.4314E+00	7.4284E+00	7.4353E+00	3.0899E-03	3.8409E-03
0.3333	8.6603E+00	8.6576E+00	8.6633E+00	2.6960E-03	3.0708E-03
0.4000	9.5106E+00	9.5106E+00	9.5106E+00	0.0000E+00	1.9073E-06
0.4667	9.9452E+00	9.9416E+00	9.9427E+00	3.5753E-03	2.4691E-03
0.5333	9.9452E+00	9.9416E+00	9.9428E+00	3.5763E-03	2.4681E-03
0.6000	9.5106E+00	9.5106E+00	9.5106E+00	0.0000E+00	9.5367E-07
0.6667	8.6603E+00	8.6576E+00	8.6633E+00	2.6970E-03	3.0699E-03
0.7333	7.4314E+00	7.4284E+00	7.4353E+00	3.0899E-03	3.8409E-03
0.8000	5.8779E+00	5.8779E+00	5.8778E+00	0.0000E+00	4.7684E-07
0.8667	4.0674E+00	4.0666E+00	4.0595E+00	7.8726E-04	7.8325E-03
0.9333	2.0791E+00	2.0777E+00	2.0660E+00	1.4238E-03	1.3128E-02
1.0000	-4.6193E-06	-4.7713E-06	-4.5644E-06	1.5202E-07	5.4946E-08

max|f-S|= 3.5762787E-03
max|f-L|= 1.3128281E-02